```
Clear["p", "m", "x", "g", "gg", "λ", "Δt", "length", "rnd", "θ"]
$Assumptions = x > 0 && x ∈ Reals && λ > 0 && λ ∈ Reals;

𝒟 = HalfNormalDistribution[θ = Sqrt[π]];
p = PDF[𝒟, x]
m = Mean[𝒟]
K2 = FullSimplify[-2 λ/p Integrate[(x - m) p, {x, 0, x}]] (*g^2[x]*)
g[x_] = Sqrt[K2]
gg[x_] = FullSimplify[Sqrt[K2] D[Sqrt[K2], x]]
```

$$\begin{cases} \frac{2 e^{-x^2}}{\sqrt{\pi}} & x > 0 \\ 0 & \text{True} \end{cases}$$

$$\frac{1}{\sqrt{\pi}}$$

$$\lambda - e^{x^2} \lambda \, \mathrm{Erfc}[x]$$

$$\sqrt{\lambda - e^{x^2} \lambda \, \mathrm{Erfc}[x]}$$

$$\lambda \left(\frac{1}{\sqrt{\pi}} - e^{x^2} x \, \mathrm{Erfc}[x]\right)$$

```
θ = Sqrt[π];
λ = 50;
Δt = 1/10000;
length = 5 × 10^6;
SeedRandom[1]
rnd = RandomVariate[NormalDistribution[], length];
ξ[n_Integer] := rnd[[n]]
vals = RecurrenceTable[
   {y[n + 1] == 1/(1 + λ Δt) (y[n] + λ m Δt + g[y[n]] Sqrt[Δt] ξ[n] + 1/2 gg[y[n]] Δt (ξ[n]^2 - 1)),
    y[1] == 1}, y, {n, 1, length}];
{Mean[vals], StandardDeviation[vals], Min[vals], Max[vals]}
```

{0.560261, 0.422413, 0.00105407, 3.32923}

```
normFactor = CovarianceFunction[vals, 0];
autocorrelationPlot = ListLinePlot[
  {ParallelTable[{z Δt 10^3, CovarianceFunction[vals, z] / normFactor}, {z, 0, 500, 25}],
   Table[{z Δt 10^3, Exp[-z Δt λ ]}, {z, 0, 500, 25}]},
  PlotRange → Full, GridLines → Automatic,
  PlotLegends → Placed[{"Simulation", "Exp[-λτ]"}, {.8, .8}],
  AxesLabel → {"τ  [msec]", "Cxx(τ)"},
  LabelStyle → Directive[Black, 13, FontFamily -> "Times New Roman"],
  TicksStyle → Directive[Black, 13, FontFamily → "Times New Roman"],
  PlotMarkers → {{Graphics[{Disk[]}], .04}, ""}]
timePlot = ListLinePlot[Thread[{Table[Δt x, {x, 1, 500}] 10^3, vals[[1 ;; 500]]}],
  LabelStyle → Directive[Black, 13, FontFamily → "Times New Roman"],
  TicksStyle → Directive[Black, 13, FontFamily → "Times New Roman"],
  AxesLabel → {"t  [msec]", "x(t)"}, GridLines → Automatic
 ]
```





```
spoint = Solve[PDF[𝒟, x] == 5 × 10^-3][[All, 1, 2]][[1]]
h1 = Histogram[vals, 75, "PDF", ChartLegends →
     Placed[SwatchLegend[{"Hist"}], Scaled[{.5, 1}]], PerformanceGoal → "Speed",
    PlotRange → {{0, Max[vals]}, {0, 1.2}}, AxesLabel → {"x", "px(x)"}];
h3 = Histogram[vals, 75, "PDF", PlotRange → {{spoint, Max[vals]}, {0, 5 × 10^-3}},
    PlotRangePadding → Automatic, PerformanceGoal → "Speed", AxesOrigin → {Max[vals] - .2, 0},
    Ticks → {Automatic, Table[{i, ScientificForm[N@i, 3]}, {i, 0, 5 × 10^-3, 5 × 10^-3 / 2}]}];
h2 = Plot[PDF[𝒟, x], {x, 0, Max[vals]},
    PlotLegends → Placed[LineLegend[{"PDF"}], Scaled[{.5, 1}]]];
h4 = Plot[PDF[𝒟, x], {x, spoint, Max[vals]}];
pdfPlot = Show[h1, h2, LabelStyle → Directive[Black, 13, FontFamily → "Times New Roman"],
   AxesLabel → {"x", "px(x)"},
   TicksStyle → Directive[Black, 13, FontFamily → "Times New Roman"],
   Epilog → Inset[Show[h3, h4], Scaled[{.8, .65}], Automatic, Automatic]]
```

$$\sqrt{\frac{1}{2}\left(8\,\mathrm{Log}[2]+4\,\mathrm{Log}[5]-\mathrm{Log}[\pi]\right)}$$



```
step = Round[1 / (λ Δt)]
tests = DistributionFitTest[vals[[1 ;; ;; step]], 𝒟, "AllTests"]
Table[ToExpression[x <> "Test"] [vals[[1 ;; ;; step]],
  𝒟, "TestConclusion", SignificanceLevel → 0.001], {x, tests}]
(*AndersonDarlingTest[vals[[1;;;;step]],𝒟,"TestConclusion",SignificanceLevel→0.001]
 CramerVonMisesTest[vals[[1;;;;step]],𝒟,"TestConclusion",SignificanceLevel→0.001]
 KolmogorovSmirnovTest[vals[[1;;;;step]],𝒟,"TestConclusion",SignificanceLevel→0.001]
 KuiperTest[vals[[1;;;;step]],𝒟,"TestConclusion",SignificanceLevel→0.001]
 PearsonChiSquareTest[vals[[1;;;;step]],𝒟,"TestConclusion",SignificanceLevel→0.001]
 WatsonUSquareTest[vals[[1;;;;step]],𝒟,"TestConclusion",SignificanceLevel→0.001]*)
probabilityPlot = ProbabilityPlot[vals[[1 ;; ;; step]], 𝒟,
  FrameLabel → {"Empirical Probabilities", "Theoretical Probabilities"},
  TicksStyle → Directive[Black, 13, FontFamily → "Times New Roman"], LabelStyle →
   Directive[Black, 13, FontFamily → "Times New Roman"], PlotTheme → "Detailed"]
```

```
200
```

```
{AndersonDarling, CramerVonMises,
 KolmogorovSmirnov, Kuiper, PearsonChiSquare, WatsonUSquare}
```

```
{The null hypothesis that
  the data is distributed according to the HalfNormalDistribution[√π]
  is not rejected at the 0.1 percent level
  based on the Anderson-Darling test., The null hypothesis that
  the data is distributed according to the HalfNormalDistribution[√π]
  is not rejected at the 0.1 percent level based on the Cramér-von Mises test.,
 The null hypothesis that the data is distributed according to the
   HalfNormalDistribution[√π] is not rejected at the 0.1 percent level
  based on the Kolmogorov-Smirnov test., The null hypothesis that
  the data is distributed according to the HalfNormalDistribution[√π]
  is not rejected at the 0.1 percent level based on the Kuiper test.,
 The null hypothesis that the data is distributed according to the
   HalfNormalDistribution[√π] is not rejected at the
  0.1 percent level based on the Pearson χ² test., The null hypothesis that
  the data is distributed according to the HalfNormalDistribution[√π]
  is not rejected at the 0.1 percent level based on the Watson U² test.}
```



```
SetDirectory[NotebookDirectory[]];
Export["autocorrelationPlotHN.pdf", autocorrelationPlot];
Export["autocorrelationPlotHN.png", autocorrelationPlot];
Export["timePlotHN.pdf", timePlot]; Export["timePlotHN.png", timePlot];
Export["pdfPlotHN.pdf", pdfPlot]; Export["pdfPlotHN.png", pdfPlot];
Export["probabilityPlotHN.pdf", probabilityPlot];
Export["probabilityPlotHN.png", probabilityPlot];
```